

The CDS-ISIS Formatting Language Made Easy

Introduction

The most complex aspect of CDS-ISIS is the formatting language, used to display, sort, print and download records. The formatting language is very flexible, and enables any number of formats to be created for many different purposes.

The complexity of the formatting language can put people off using the software. Therefore the best way to learn the formatting language, is to build up formats slowly, learning each step at a time.

Not all aspects of display/print formats are discussed here. However once skills have been developed to the levels discussed here, it is time to use the UNESCO manual for the more advanced aspects.

Before starting

It is important either to know all the numerical tags assigned to each field very well, or to use the print screen button to printout a list of the tags, sub-fields and repeatable fields from the Field Definition Table (FDT).

CDS-ISIS Field Definition Table

Select the relevant database by pressing **C** for Change database from the main menu, and then giving the name of the database.

Select **D** for Define database services, then **U** for Modify database definitions, and then **A** for Update field definition table.

Tag	Name	Len	Typ	Rep	Delimi
300	Personal authors	1000	X	R	
310	Corporate author	1000	X	R	ab
320	Meeting	1000	X		abgij
330	Author's affiliation	400	X		ab
400	Place publicat				
440	Date of publ				
450	Serial number				
460	Physical des				
480	Series				
490	Part stateme				

Field Definition Tables (FDT)

Tag	Name	Type	Rep	Pattern/Subfiel
300	Personal authors	Alphanumeric	X	
300	Personal authors	Alphanumer	R	
310	Corporate author	Alphanumer	R	ab
400	Place publication/publisher	Alphanumer	R	ab
440	Date of publication	Alphanumer	-	b
460	Physical description	Alphanumer	-	abcd
480	Series	Alphanumer	R	abc
500	Note	Alphanumer	R	
610	Classification/Location	Alphanumer	R	acd
620	Descriptor	Alphanumer	R	ab
630	Geographic descriptors	Alphanumer	R	

WINISIS Field Definition Table

Select Field Definition Table from the Edit menu.

Editing and Saving Display/Print Formats using CDS-ISIS for DOS

Editing and browsing formats

Select **S** from the Main menu

Select **F** to change the display format

If the current format is to be edited, begin editing the format.

If a new format is to be designed, press **F6** to clear the current format, then type the name of the new format, preceded by an **@** symbol.

```
Please enter/edit format (@xxxxx to use predefined format)
@print
```

Changes to the display format can be made here, and the result viewed by pressing **<Enter>** followed by **B** for Browse Masterfile. Press **F** to return to the format.

View the format using browse each time a small change is made. This way if the format does not work, and an error message appears, it is easy to check whether something is missing, for example forgetting to close an opened bracket, or end quotation marks etc. If it is not possible to see what is wrong, remove all the latest changes and try again.

Press **F8** to save the format, when changes have been made that improve the format

Selecting formats

To select a particular format press **F6** to clear the current format, and type **@** plus the name of the format to be edited, for example **@print**

Copying and deleting formats

When a good format exists, and a new part of the formatting language is to be tried, it is sensible to make a copy of the format, so that the current useful format is still saved. Several different formats can be saved, until all the formats needed have been created. It is possible to delete unwanted formats, so do not worry about ending up with too many confusing formats.

Select **D** for Database Definition Services from the Main menu

Select **U** for Modify Database Definition

Select **K** for List Database Parameter Files

```
Files defined for Data Base NDC

Worksheet Descriptions
  1.NDC

Display formats
  2.NDC      3.NDC2      4.NDC3

Field Select Tables
  5.NDC      6.NDC2      7.CF-ND      8.ND-CF
```

Files defined for a specific database using CDS-ISIS

The display/print formats are listed and numbered. To copy a format, select **C** for copy and give the number next to the name of the format to be copied. Give a name for the copied file, and press **<Enter>**

To delete a format, select **D** for delete, and give the number next to the name of the format to be deleted. Before deleting a format, first check that it is a format that is not needed, by using it to browse the database.

To edit copied formats, return to the Main menu by selecting **X** at each menu. Then select **S** for the Information Retrieval Services menu.

Editing and Saving Display/Print Formats using WINISIS

To edit a format click the split/unsplit view button



Then click on the Edit PFT button on the toolbar.

When a change has been made, the change can be display by clicking on the change button (lightening symbol). This does not save the change, so if not happy with the results, the previous format can be brought back by selecting a different format, then re-selecting the original format. Once happy with a format, click on the save button.



It is also possible to edit formats by selecting Print formats from the Edit menu. This has some useful automated functions for adding bold, italic or underline to the text. However the changes cannot be viewed without saving the format.

Creating Formats


Listing each field on a separate line

Type the tags of each field in the required order: e.g. **300 200**

Place a **V** in front of each tag number to identify it as a tag number rather than just a number: e.g. **v300 v200**

Place a comma between the statements in addition to the space. Spaces are not recognised by the formatting language and only serve to make the statements easier to follow by the person looking at the screen. Commas are recognised as separating/grouping punctuation by the program, in the same way as they are in non-computer language: e.g. **v300, v200,**


Type **/** between the tag number and the comma. This will place each field on a separate line, and will make it easier to view the information before a more advanced display format is designed: e.g. **v300/, v200/,**




Modes

When developing print/display formats, the default mode is Proof mode. This means that all the indexing and sorting marks such as < > and //, and the sub-field markers ^a will appear, although the repeatable field markers % will not appear, and the repeatable text will be squashed together without any spaces.

In order to avoid this happening, it is important to put a statement at the beginning of the display/print format that tell ISIS that you want heading mode, therefore add the following to the beginning of the format: **mhl**,



If wanting to change a piece of text (for example authors or descriptors) into capital letters, use **,mhu**, instead. However remember to put **mhl** again when you want the text to return to the case in which it was entered.



Adding Text, Punctuation and Spacing


The adding of text to the information stored in the database can be useful, either to label fields for those viewing records on the screen, or to create letters for ordering materials, or recalling of overdue loans, or producing printed bibliographies.

Punctuation should never be added to the end of a field during data entry, only between elements of a field, for example surname and first name of an author, or title and subtitle where the field is not sub-fielded. Punctuation should be added as part of the display/print format, using the methods demonstrated below.

Space between fields is added in the same manner as other punctuation. There is no automatic spacing between fields when they displayed on the screen (except between sub-fields in some modes), therefore this needs to be added in the same manner as text and punctuation. Added space can appear confusing in the print format, therefore using commas to keep statements together can make the format easier to understand. The position of commas can be very important to how the format is interpreted by the program as illustrated later on.

Non-repeatable fields

The easiest way to add text/punctuation/spacing is the place it between single quotation marks, either before or after the field to be affected, as appropriate:



However fields may not always contain information, and the added text will appear, even if there is no content. This can be useful when designing a format for checking the accuracy of data entry, i.e. it will identify empty fields, for which there should be information:

In cases where it is not certain that the field will contain data, double quotation marks are used instead. The text will then only appear if there is data in the field:


The image shows a screenshot of a library catalog interface with three overlapping windows. The top window, labeled 'MFN 514', displays the title 'Computer training for teachers of secondary school students'. The middle window, labeled 'MFN 1', displays the author 'Makawa, Judith' and the title 'Computers, a trainers' handbook'. The bottom window shows a MARC record snippet: 'mhl,\"Author: \" v300,/Title: ' v200/'. The interface includes navigation buttons and a 'Format: TEST' dropdown.

Notice the punctuation that appears between the main title and the subtitle in the example above. This is the text that is automatically inserted between sub-fields **a** and **b** in the heading mode. See the section on sub-fields on how to alter this punctuation.

Position of commas


It is important to remember the commas, and to be careful where they appear in relation to the format. Try out the following example:

Mhl, "Author: " v300, "." / "Title: " v200/



By moving the comma to after the punctuation code as follows, the full stop only appears if an author is present:

Mhl, "Author: " v300". " / "Title: " v200/



Repeatable fields

If the field is repeatable (for example author, keywords etc.), the symbol | is used instead. Any text appearing between these symbols will automatically appear for each repeat of the field, for example

v620|/|, or **|/|v620**,

Note that the position of the added text in relation to the field marker, effects where the text appears in the printout or display.


The image shows two overlapping windows from a software application, likely a library catalog or database viewer. Both windows have a header bar with a field identifier, navigation buttons (left, right, double left, double right), and a 'Format: TEST' dropdown menu.

The top window is labeled 'MFN 1'. It displays the following text:
Author: Makawa, Judith.
Title: Computers, a trainers' handbook
Subject: computers / training /
Below the text is a command line: `mhl,"Author: "v300". "/"Title: "v200/,"Subject: "v620|/|/`

The bottom window is labeled 'MFN 514'. It displays the following text:
Title: Computer training for teachers of secondary school students
Subject: / computers / training / schools
Below the text is a command line: `mhl,"Author: "v300". "/"Title: "v200/,"Subject: "|/|v620/`


Using the method above, the added text appears before or after every occurrence of the repeatable field. In order to remove the unnecessary extra text, the following is used instead.

v620+|/|, or **|/|+v620**,



Note that both examples look the same. However, the first example says: print the / after each occurrence of the field except the last. The second example says: print the / before each occurrence of the field except the first. The best format to use will depend on what field is displayed/printed before of after this field.


|/|+V620,|/|v630,



The best way to decide which format to use, is to try it out both ways and look at the results.


Fields with Sub-fields

Fields with sub-fields have to be dealt with slightly differently. If using heading mode, CDS-ISIS will add its own punctuation between sub-fields, however this punctuation is not always what is wanted. In the example below, CDS-ISIS places a comma between the title and the subtitle, and the place of publication and the publisher. In both cases the fields are subdivided into sub-fields **a** and **b**.



Therefore it is best to list each sub-field and the relevant punctuation and spacing:

<i>Format</i>	<i>Result</i>
V400	London, Library Association
V400^a' : ', v400^b	London : Library Association
v400^a' : ', v400^b', ', V440,	London : Library Association, 1999



When working with sub-fields, it is important that for fields expected to contain data (as above) the single quotation mark is used, as this will help to indicate where information is missing from a record.

When working with fields that do not always include information, such as a subtitle, to use double quotation marks, so that the punctuation only prints if a sub-field exists. Also note that for the title statement, the comma comes before the quotation marks. This ensures that the punctuation is only printed if a sub-field is present:

<i>Format</i>	<i>Result</i>
v200^a, " : "v200^b,	Computers : a trainer's handbook
v200^a, " : "v200^b,	Computer training for teachers
v200^a" : "v200^b,	Computer training for teachers :

Page Layout


New lines and pages

In all the formats, the fields have been put on separate lines using the format to start a new line:

/ = start a new line (on condition the previous line is not blank)

However in order to put each occurrence of a repeatable field on a new line, the format is slightly different, a repeating field needs to be enclosed in brackets to ensure that all parts of the repeating field are effected:

(V300/)



It is also possible to start a new line even if the previous line is blank - useful at the end of a format to ensure blank lines between each record:

= start a new line under any condition.

For print formats, it is possible to start a new line using **NP**. This will continue the printout on the next page.


To continue on the next page, only if there is less than a certain number of lines left on the current page, use **NP(4)** = start a new page if less than 4 lines are left on the current page.

Indenting

It is possible to indent text, indicating the position on the first and following lines using:

(n,n) = indentation positions for first and following lines


If the following is placed after the field number **(3,6)** then the first line will start 3 spaces in, and all following lines 6 spaces in from the margin. However, if the first line is preceded by text between ' ' or " " the first line will start immediately after this text, and not at the stated position.



Columns and Limiting of text length


Text can be placed in columns, which will not look impressive when displayed on the screen, but can be very useful when printing. The following code is used in combination with the number of the position on the screen/page:

C = start at a stated position - e.g. **C5** = start at position 5 on the page.



Note that when using column display, that if the text in a column is too long, the text in the next column will print on the following line. In the first example above, only sub-field **a** of the title is displayed, this being so short that the columns are correctly displayed. However, it is possible to limit the length of text displayed/printed for each field, using **.n**


e.g. **v200.20** will limit the length of the title that prints/displays to 20 characters.



It is also possible to start the display/printout from a position other than the first character in the field. For example, if the date field includes the month as well as the year, this may take up too much room in a columnar display. It is possible to only print the information necessary.

This is done using ***n**

By placing ***2** immediately after the field tag, the first 2 characters of the field are not displayed/printed



Proof reading format

If creating a format for proof reading, use the proof mode, by putting **,mpl**, at the beginning of the format. This ensures that the < > and the //, and the ^a appear on the display/printout, in order to check correct usage.

Add labels for all fields to help identify any data that has been put in the wrong fields. For fields that **MUST** be present in a record, use a single quote mark ' to surround the text, for non-essential fields use **double quote marks** " to surround text. For repeatable fields use repeatable field markers as part of the text

MFN 1 Format: TEST

Author: Makawa, Judith
Title: ^aComputers^ba trainers' handbook
Place/Publisher: ^aLondon^bLibrary Association
Date: 1999
Subject: computerstraining
Geographic: BangladeshIndia

mpl, "Author: " v300 , / ' Title: ' v200 , / ' Place/Publisher: ' v400 , / ' Date: ' v440 ,

In order to tidy up the display/printout and start each line at the same position, extra spaces can be inserted within the quotation marks, so that each field label is of the same length. For example label **Place/Publisher:** is 18 characters long, therefore 10 spaces need to be added after **Author:**

MFN 1 Format: TEST

Author: Makawa, Judith
Title: ^aComputers^ba trainers' handbook
Place/Publisher: ^aLondon^bLibrary Association
Date: 1999
Subject: computers / training
Geographic: Bangladesh / India

mpl, "Author: " v300 , / ' Title: ' v200 , / ' Place/Publisher: ' v400 ,

More Advanced Formats

There will be times when more complex print formats are required, for example when wanting to display/print records in catalogue card style, or when producing bibliographies or resource lists.

The formatting used in more complex display/print formats may depend on the presence or absence of certain fields, for example a record without an author may need to be presented in a different manner than a record with an author etc.

The way to tell ISIS to see if a field is present is to say **p(v300)** = an author is present in this particular record, or **a(v300)** = there is no author for this particular record.

This is the simple way to tell ISIS to look for situations:


if this is the case **then** do this and **finish** the statement

Statement 1

If there are any authors for this record, **then** put them first, and put the title indented on the next line, **finish** the statement

Format

if p(v300) then v300(3,3) / , v200^a , " : " v200^b fi




Statement 2

If there are no authors for this record, **then** put the title first, with the second line indented, **then** finish the statement

Format

if a(v300) then v200^a(3,3) , " : " v200^b(3,3) fi



In the first example above the record without an author displays as a blank, and in the second, the record with an author displays as a blank. By putting both statements into the format, then the record will appear whether there is an author or not. However, there is another way of telling ISIS to look for these situations. Although it is more complex, it cuts down on the amount of different statements that have to be made:


if this is the case **then** do this **else** (otherwise) do something else and **finish** the first statement and **finish** the second statement

Statement

If there are authors for this record, **then** put them first, and put title on the next line, with the first line indented, **else if** (otherwise if) there are no authors, put the title first, with the second line indented, **finish** the first statement and **finish** the second statement

Format

if p(v300) then v300 (3,3) , / v200^a, " : " v200^b, else if a(v300) then v200^a(3,3) , " : " v200^b(3,3), fi fi



More complex displays should be build up gradually, and each stage should be checked to see if the format is working. If the display will not appear on the screen and an error message appears, see if there is anything missing – for example forgetting to closing an open bracket or close quotation marks, forgetting to put **fi** or **fi fi** at the end of the statement etc. If the format still does not work, delete the latest changes made, and try again.

Making the format look better

It is possible to add formatting such as bold, underline, or italics to make certain text stand out.


b, = bold *i*, = italics u, = underline

Place the text to be affected between { }

For example:

{b,V300}, *{i, V200}*

{u,{b,V300}}, *{i, V200}*



It is also possible to use different fonts and font sizes, see the CDS-ISIS or WINISIS manual for more details.

Advanced formats available to WINISIS only

With WINISIS it is possible to open documents (word processed files) and web pages. It is also possible to include graphics as part of a display format.

The format is:

mpl, LINK ((‘description’), ‘CMD program’, field)


The *description* is what is displayed on the screen. The displayed description can be clicked to open the file or web site, or to display the image.

The *program* is the name of the program required, for example Netscape, Word, or Paintbrush

The *field* is that which contains the URL, or file name and path for the word processed file or graphic.

Format

mpl, LINK ((‘Datahome Website’), ‘CMD Netscape’, v998)



By clicking on the description, the web site, or document will be opened.

It is possible to include both the description and the program information into sub-fields of the field containing the website or document number. This enables the format able to cope with a variety of different programs and descriptions. For example:

998^a contains the information about a document or web site related to the record

998^b contains the description to appear on the screen

998^c contains the name of the program required to open the document or web site

The following format would cover all requirements for all records with links to documents or web sites:

mpl, LINK (('v998^b'), 'CMD v998^c', v998^a)

There are additional formats that are only available to WINISIS, some of which update those available to CDS-ISIS. See the WINSIS manual for details